

Connecting the Dots: Using ODS RTF ANCHORS, ODS RTF BOOKMARKS and Table of Contents to Guide Readers Through Your SAS® RTF Reports

Debra A. Goldman, MS, Memorial Sloan Kettering Cancer Center

ABSTRACT

In today's world, many of us send our reports via email, sometimes never meeting a client in person. Lengthy reports or dense results may be difficult to follow or even overwhelming for clients, even if we've provided everything they've asked for (and more!). The number of results pages may be uncontrollable, but fortunately, the organization and display of findings are controllable. The use of table of contents (TOC) and hyperlinks helps guide clients through the work we've completed by providing structure and connecting related findings.

This paper will demonstrate the TOC feature, and the use of ODS RTF ANCHORS and ODS RTF BOOKMARKS within PROC ODSTEXT to guide users through one's RTF report in SAS 9.4. The relative merits and shortcomings of each method will be presented, along with case use examples. These methods are applicable across disciplines, but the examples will come from biostatistics. Attendees should have a basic understanding of the ODS RTF destination and PROC ODSTEXT.

INTRODUCTION

When we are taught to write essays, we learn that the general format is introduction, body of paper, and conclusion. When we submit a SAS paper along with our WUSS presentation, we are provided with a template of how to organize a SAS paper, which includes an abstract summary and an introduction. When we present, we start with a slide that tells the attendees what the presentation will cover. Yet, when we produce reports ourselves, we tend to jump right into the body with a methods section, a Table 1, or even the main findings without any introduction to the reports' contents.

Introducing a report with a Table of Contents (TOC) or an executive summary provides the structure of the analyses and eases readers into a content of a report. It tells readers what to expect. Further, with the features of ODS RTF, one can connect findings and provide an interactive way for readers to follow along.

This paper will demonstrate how to:

- Create an automatic TOC and label PROC within SAS
- Output a basic list using PROC ODSTEXT
- Add hyperlinks to create a customized TOC
- Add hyperlinks throughout the report to connect different findings

As the automatic TOC has been covered extensively in other papers, this paper will mainly focus on creating a manual TOC with PROC ODS TEXT. I will be using the SASHELP dataset, SASHELP.HEART, throughout this paper. This paper is intended for those who use the ODS RTF destination and have experience with PROC FREQ, PROC REPORT, and PROC SGPLOT.

EXAMPLE DATASET

DATASET #1: SASHELP.HEART

SASHELP.HEART contains data on 5,209 patients from the Framingham Heart Study. Please see SASHELP documentation for more information.¹

Alphabetic List of Variables and Attributes				
#	Variable	Type	Len	Label
12	AgeAtDeath	Num	8	Age at Death
5	AgeAtStart	Num	8	Age at Start
3	AgeCHDdiag	Num	8	Age CHD Diagnosed
15	BP_Status	Char	7	Blood Pressure Status
14	Chol_Status	Char	10	Cholesterol Status
13	Cholesterol	Num	8	
2	DeathCause	Char	26	Cause of Death
8	Diastolic	Num	8	
6	Height	Num	8	
10	MRW	Num	8	Metropolitan Relative Weight
4	Sex	Char	6	
11	Smoking	Num	8	
17	Smoking_Status	Char	17	Smoking Status
1	Status	Char	5	
9	Systolic	Num	8	
7	Weight	Num	8	
16	Weight_Status	Char	11	Weight Status

Output 1. PROC CONTENTS of SASHELP.HEART

AUTOMATIC TABLE OF CONTENTS

The ability to insert a table of contents in an ODS RTF document using SAS has been around at least since 2004 and has been demonstrated at various SAS conferences²⁻⁶. Consequently, I will touch on this method in brief and refer readers to the references for a more detailed description.

Step 1: Add statements to ODS RTF command

The key to including a TOC is two simple statements: `contents =yes` and `toc_data`. See below for the inclusion of these statements within the ODS RTF statement.

```
options orientation="Portrait";
ods graphics on /width=5in height=4in; ods listing close;
ods rtf file="G:\My Documents\Biostatistics\WUSS\2018\StatisticalAnalyses\SAS
Output\WUSS_ExampleReport_V1_30July18.rtf"
style=wuss2018 headery=320 startpage=no contents =yes toc_data; title; run;
```

Step 2: LABEL PROC within SAS

Next, label each PROC. These will show up in the TOC within Microsoft Word. Some procedures also allow labeling of the sub-contents:

```
ods proclabel='Table 1. Patient Gender';  
proc freq data=sashelp.heart;  
tables sex/nocum plots=none;  
run;
```

Step 3: Run Report and Perform Commands in Microsoft Word

After you've run the report, highlight all the contents. This can be done through the command **CTRL + A** [Select All] and then press **F9**. The first command selects all content in the document and the second updates all TOC indices within the Microsoft Word document.

Table of Contents	
Table 1: Patient Gender.....	1
Table Sex.....	1
One-Way Frequencies.....	1

Output 2. Output from TOC Command, CTRL+A and F9

Full Code:

```
options orientation="Portrait";  
ods graphics on /width=5in height=4in; ods listing close;  
ods rtf file="G:\My Documents\Biostatistics\WUSS\2018\StatisticalAnalyses\SAS  
Output\WUSS_ExampleReport_V1_30July18.rtf"  
style=wuss2018 headery=320 startpage=no contents =yes toc_data; title; run;  
ods escapechar="^";  
options missing=" ";  
title;  
ods noproctitle;  
ods text="WUSS 2018 Output Report  
^n Full Report - Version 1.0  
^n Debra A. Goldman, MS  
^n Summer 2018 ^n";  
  
ods proclabel='Table 1. Patient Gender';  
proc freq data=sashelp.heart;  
tables sex/nocum plots=none;  
run;  
ods rtf close;
```

Table of Contents		
Table 1: Patient Gender		1
Table Sex		1
One-Way Frequencies		1
WUSS 2018 Output Report		
Full Report - Version 1.0		
Debra A. Goldman, MS		
Summer 2018		

Sex	Frequency	Percent
Female	2873	55.15
Male	2336	44.85

Output 3. Full Output for Automatic TOC

As one can see, the TOC appears above the initial header, in a different font (Albany AMT vs. Arial), individually spaced, and contains three different headers' information. While this can be updated if one manually moves the table to another page, it requires subsequent manual steps in word to adjust the appearance and formatting. Below are the pros and cons to using the above method:

PROS AND CONS:

PROS:	CONS:
<ul style="list-style-type: none"> Automatically provides full list of PROC performed and tables output 	<ul style="list-style-type: none"> Requires manual command to add
<ul style="list-style-type: none"> Page numbers can be updated with use of F9 	<ul style="list-style-type: none"> Automatic additional labeling of contents
	<ul style="list-style-type: none"> Not double spaced
	<ul style="list-style-type: none"> Formatting differs from other ODS text; needs to be modified in word
	<ul style="list-style-type: none"> Provides too many details
	<ul style="list-style-type: none"> Requires manual deletion

Table 1. Pros and Cons of using automatic TOC

This method doesn't work well if one tries to avoid any additional steps outside SAS that need to be repeated if the report is re-generated. Therefore, I choose to use a combination of the options below.

MANUAL TABLE OF CONTENTS WITH PROC ODSTEXT AND ODS RTF ANCHORS

As mentioned in the introduction, this paper will mainly focus on the manual method for creating a TOC. The first step is to use SAS's versatile procedure, PROC ODSTEXT.

INTRODUCTION TO PROC ODSTEXT

As discussed in a prior WUSS paper⁷, PROC ODSTEXT is a powerful tool to insert text into RTF reports. The prior report focused on creation of paragraphs, while this paper will focus on lists.

The `list` command in PROC ODSTEXT tells SAS to make a list⁸. Each item corresponds to a new row. The commands within the style provide color, change fonts, and allow us to add internal and external hyperlinks. In addition to the `list` command, an `end` statement is needed to stop the list. Multiple lists can be included in the same PROC ODSTEXT statement.

Below is the sample code for creating a list:

```
proc odstext;
p "This is a sample list" / style=[fontsize=11pt fontweight=bold
fontfamily=Arial];
list / style={liststyletype="decimal" fontsize=11pt};
item "This is the first line of the list ^n ^n" / style=[fontsize=11pt
fontfamily=Times color=green url='#FirstLine'];
item "This is the second line ^n ^n" / style=[fontsize=10pt fontfamily=Arial
color=blue url='#SecondLine'];
item "This is the third line of the list ^n ^n" / style=[fontsize=12pt
fontfamily=Courier color=red url='#ThirdLine'];
end;
run;
```

SAS Syntax	RTF/SAS Function
List	Tells SAS to create a list
liststyletype="decimal"	Creates a numbered list
url='#FirstLine'	Points to an internal hyperlink called "FirstLine"
^n ^n	Double spaces a line. Each refers to a carriage return
end;	Ends list

Table 2. Explanation of PROC ODSTEXT commands

The additional syntax above in the `style=` changes the formatting of the text in the item line. Below is the output:

```
This is a sample list
  1. This is the first line of the list

  2. This is the second line

  3. This is the third line of the list
```

Output 4. Sample list using PROC ODSTEXT

Please see the recommended reading for more information on using PROC ODSTEXT.

CREATING A MANUAL TABLE OF CONTENTS IN PROC ODSTEXT

Step 1: Create list with URL for each relevant PROC

The first step in creating a TOC is to make a list like the one above:

```
proc odstext;
p "Table of Contents" / style=[fontsize=11pt fontweight=bold
fontfamily=Arial];
list / style={liststyletype="decimal" fontsize=11pt};
item "Statistical Methods ^n ^n" / style=[fontsize=11pt fontfamily=Arial
color=blue url='#Methods'];
item "Patient and Baseline Descriptive Characteristics Overall and by Gender
^n ^n" / style=[fontsize=11pt fontfamily=Arial color=blue url='#Table1'];
item "Association between Age and Cholesterol ^n ^n" / style=[fontsize=11pt
fontfamily=Arial color=blue url='#Table2'];
```

```

item "Boxplot of Age by Cholesterol Status ^n ^n"/style=[fontsize=11pt
fontfamily=Arial color=blue url='#Figure1'];
end;
run;

```

Table of Contents
1. Statistical Methods
2. Patient and Baseline Descriptive Characteristics Overall and by Gender
3. Association between Age and Cholesterol
4. Boxplot of Age by Cholesterol Status

Output 5. Table of Contents List

Step 2: Use ODS RTF ANCHOR to label each PROC

Next, each procedure should have a preceding statement that matches the URL, excluding the #:

```

ods rtf anchor="Table1";
ods text="^S={fontweight=bold}Table 1. Patient Characteristics for Framingham
Heart Study";
proc report data=table1_patientchar nowindows headline missing spanrows;
...

```

```

ods rtf anchor="Figure1";
ods text="^S={fontweight=bold}Figure 1. Box plot of Age at Start by
Cholesterol Status for Each Gender";
proc sgplot data=sashelp.heart;
...

```

SAS Syntax	RTF/SAS Function
ods rtf anchor=	Creates internal link anchor in document that shows URL where to move to

Table 3. Explanation of anchor statements

*Importantly, RTF anchor statements must follow the URL call. If the URL call is after the anchor statement, the link **will not work**.*

Step 3: Guide users with links in PROC ODSTEXT and in PROC REPORT statements

URLs can also be placed within PROC ODSTEXT statements or in procedures, such as PROC REPORT. These can be used to point readers to related findings. Below are three places where URL can be included:

```

proc report data=table_chol_age nowindows headline missing spanrows;
column var1_name ("Gender" sex) (variable2,combo) ("p-value" pval dummyvar);
define var1_name/group order=data '' format=$desc_variablef.;
define sex/group order=data '';
define variable2/across 'Cholesterol Status ^n Median (Range)' order=data
style=[url='#Figure1'];

```

```

define combo/display '' style=[textalign=center cellwidth=1.2in];
define pval/' ' sum format=pval_32p. style=[fontweight=pval_fw.];
define dummyvar/computed noprint ;
compute dummyvar;
dummyvar=1;
endcomp;
where combo ne " " and variable2 ne "Unknown";
compute after sex;
  length text $ 150;
  if sex="Female" then do;
    text = "^S={textalign=left just=left fontstyle=italic fontsize=9pt
url='#Figure1'}Click here for associated plot";
    num=150;
  end;
  else do;
    text = "";
    num=0;
  end;
  line text $varying. num ;
endcomp;
run;

proc odstext;
p "Please see next page for plot of age distribution for each cholesterol
grouping" / style=[color=blue fontstyle=italic url='Figure1'];
run;

```

	Gender	Cholesterol Status Median (Range)			p-value
		Desirable	Borderline	High	
Age at Study Start, years	Male	42 (29-62)	43 (29-62)	45 (29-61)	<.001
	Female	37 (29-62)	43 (29-62)	49 (28-62)	<.001
<i>Click here for associated plot</i>					
<i>Please see next page for plot of age distribution for each cholesterol grouping</i>					

Output 6. PROC REPORT and PROC ODSTEXT Output

As above, URL statements can be placed in any `style=` statement to create a hyperlink.

PROS AND CONS:

PROS:	CONS:
<ul style="list-style-type: none">• Customizable in SAS	<ul style="list-style-type: none">• Manual page number insertion (if relevant)
<ul style="list-style-type: none">• Completely reproducible (no work needed in word)	<ul style="list-style-type: none">• Subject to typos and incorrect link specification
	<ul style="list-style-type: none">• Can only use links to move forwards in document

Table 4. Pros and Cons of using PROC ODSTEXT and RTF ANCHORS

Although this output allows for a great deal of customization and no work outside SAS, there's still room for improvement. The links aren't created automatically and are subject to typos or incorrect specification. Also, page numbers are not included, which may be important to some users, and one can only move in a single direction. The final method solves the problem of direction and allows for page numbers to be included.

MANUAL TABLE OF CONTENTS WITH PROC ODSTEXT AND ODS RTF BOOKMARKS

ODS RTF ANCHORS are a great start to manual guidance. As illustrated above, they can be placed in PROC ODSTEXT or within PROC REPORT. However, they can only move users from beginning to end. They cannot move users to the beginning or to connect relevant findings. Two ways to insert bookmarks that allow users to move throughout the document are presented below.

OPTION 1: USE RTF CODE WORDS TO CREATE AND CALL PROCEDURES

RTF language can be inserted into single or double quotes and will not print out with the use of `*\` syntax preceding the RTF code word. As this language will remain constant regardless of software used, this method is beneficial to learn if one also uses R or other LaTeX systems.

Step 1: Create list with URL for each relevant PROC

The use of URL works the same for bookmarks as it does for RTF anchors, so Step 1 here is identical to Step 1 above.

Step 2: Use RTF code words to label each PROC

The statements, `*\bkmkstart`, and, `{*\bkmkend`, appear within the paragraph in PROC ODSTEXT:

```
proc odstext;  
p "{\*\bkmkstart home}Table of Contents{\*\bkmkend home}" /  
style=[fontsize=11pt fontweight=bold fontfamily=Arial];  
list / style={liststyletype="decimal" fontsize=11pt};  
item "Statistical Methods ^n ^n"/style=[fontsize=11pt fontfamily=Arial  
color=blue url='#Methods'];  
...  
run;
```

These can be added to bookmark specific text statements, and may be particularly helpful when wanting to point to different places within the same procedure. The name of the bookmark should be listed after each of the RTF code words. The display words should appear in between the two brackets.

SAS Syntax	RTF/SAS Function
<code>*\bkmkstart</code>	Points to the start the bookmark. The name of the bookmark should follow this statement
<code>*\bmkkend</code>	Ends the call for the bookmark. The name of the bookmark should follow

Table 5. Explanation of RTF bookmark syntax

Step 3: Guide users with links throughout document

For bookmarks, URL statements can appear before or after the bookmark call. In addition to using URL statements as above, one can call RTF bookmarks with the use of the RTF code word `HYPERLINK`.

The referenced text or procedure should contain the bookmark statement, and the text that points to the bookmark should contain the field statement⁹. To call these properly, double quotes should be used around the field statement so that single quotes can be used to refer to the bookmark name:

```
proc odstext;
p "{\field {\*\fldinst HYPERLINK \l 'home'}{\fldrslt Back to Top}} ";
run;
```

SAS Syntax	RTF/SAS Function
<code>\field</code>	Tells RTF that you're referring to a field
<code>*\fldinst HYPERLINK \l</code>	Tells RTF to insert a hyperlink with the given name that follows
<code>\fldrslt</code>	Outputs results of the field call and allows text to follow

Table 4. Explanation of bookmark syntax

```
ods rtf bookmark="TOC";
proc odstext;
p "Table of Contents" / style=[fontsize=11pt fontweight=bold
fontfamily=Arial];
...
proc odstext;
p "Back to beginning" /style=[fontstyle=italic url="#TOC"];
run;
```

Backto Top
<i>Backto beginning</i>

Output 7. Two methods for referring to bookmarks

The next option is strikingly similar to ODS RTF ANCHOR, but allows users to move throughout the document.

OPTION 2: USE ODS RTF BOOKMARK TO CREATE AND CALL PROCEDURES

With respect to functionality, these two bookmark options operate identically within RTF documents.

Step 1: Create list with URL for each relevant PROCEDURE

As above, URLs work for both bookmarks and anchors. No changes need to be made to the PROC ODSTEXT

Step 2: Use ODS RTF bookmark statements to label each PROC

Instead of using the `ANCHOR` call, one simply uses the `BOOKMARK` call:

```
ods rtf bookmark="Table1";
ods text="^S={fontweight=bold}Table 1. Patient Characteristics for Framingham
Heart Study";
proc report data=table1_patientchar nowindows headline missing spanrows;
...
```

```
ods rtf bookmark="Figure1";
ods text="^S={fontweight=bold}Figure 1. Box plot of Age at Start by
Cholesterol Status for Each Gender";
proc sgplot data=sashelp.heart;
...
```

SAS Syntax	RTF/SAS Function
<code>ods rtf bookmark=</code>	Creates internal link bookmark in document for the procedure that follows

Table 5. Explanation of ODS RTF bookmark statement

Each bookmark statement must contain a unique name to work correctly.

Step 3: Guide users with links in PROC ODSTEXT and in PROC REPORT statements

Step 3 here is identical to step 3 for option 1 above. Bookmark internal links will work within PROC REPORT and PROC ODSTEXT.

Step 4: Bring users back to beginning or have them move freely about the cabin

Unlike ANCHORS, BOOKMARKS can be called in any order throughout the RTF document:

```
ods text="^S={fontweight=bold}Table 1. Patient Characteristics for Framingham
Heart Study";
ods rtf bookmark="Table1";
proc report data=table1_patientchar nowindows headline missing spanrows;
...
proc odstext;
p "{\field {\*\fldinst HYPERLINK \l 'home'}{\fldrslt Back to Top}} ";
run;

ods rtf bookmark="Table2";
ods text="^S={fontweight=bold}Table 2. Association between Cholesterol Status
and Age";
proc report data=table_chol_age nowindows headline missing spanrows;
...
proc odstext;
p "Please see next page for plot of age distribution for each cholesterol
grouping" / style=[color=blue fontstyle=italic url='Figure1'];
run;

ods startpage=now;

ods rtf bookmark="Figure1";
ods text="^S={fontweight=bold}Figure 1. Box plot of Age at Start by
Cholesterol Status for Each Gender";
proc sgplot data=sashelp.heart;
vbox ageatstart /category=chol_status displaystats=(q1 median mean q3 min
max) group=sex;
```

```

where chol_status ne " ";
xaxis label = "Cholesterol Status";
yaxis min=20 max = 70;
run;
proc odstext;
p "{\field {\*\fldinst HYPERLINK \l1 'Table2'}}{\fldrslt Click here to go back
to statistical test results}} " /style=[color=blue fontstyle=italic];
run;

```

Table 1. Patient Characteristics for Framingham Heart Study

		All Patients N (%)	Gender	
			Female N (%)	Male N (%)
Age at Study Start, years	Median (range)	43 (28-62)	43 (28-62)	44 (29-62)
Cholesterol Grouping	Desirable	1405 (27)	805 (28)	600 (25.7)
	Borderline	1861 (35.7)	959 (33.4)	902 (38.6)
	High	1791 (34.4)	1010 (35.2)	781 (33.4)
	Unknown	152 (2.9)	99 (3.4)	53 (2.3)
Height, in	Median (range)	65 (52-77)	63 (52-71)	68 (56-77)
Weight, lbs	Median (range)	150 (67-300)	138 (67-300)	167 (99-276)
Smoking Status	Non-Smoker	2501 (48)	1682 (58.5)	819 (35.1)
	Light (1-5)	579 (11.1)	422 (14.7)	157 (6.7)
	Moderate (6-15)	576 (11.1)	340 (11.8)	236 (10.1)
	Very Heavy (> 25)	471 (9)	73 (2.5)	398 (17)
	Heavy (16-25)	1046 (20.1)	339 (11.8)	707 (30.3)
	Unknown	36 (0.7)	17 (0.6)	19 (0.8)

[Back to Top](#)


Table 2. Association between Cholesterol Status and Age

		Cholesterol Status Median (Range)			p-value
	Gender	Desirable	Borderline	High	
Age at Study Start, years	Male	42 (29-62)	43 (29-62)	45 (29-61)	<.001
	Female	37 (29-62)	43 (29-62)	49 (28-62)	<.001

Click here for associated plot

Please see next page for plot of age distribution for each cholesterol grouping

Figure 1. Box plot of Age at Start by Cholesterol Status for Each Gender



[Click here to go back to statistical test results](#)

Output 8. Demonstration of various internal hyperlink methods

INSERT PAGE NUMBERS FOR BOOKMARKS

For either bookmark option, Microsoft word creates an index that stores the page number. This page number can be called within PROC ODSTEXT similar to how a hyperlink is called in RTF. Importantly, the text preceding the field reference **must** be enclosed in brackets, {}:

```
ods rtf bookmark="TOC";
proc odstext;
p "Table of Contents" / style=[fontsize=11pt fontweight=bold
fontfamily=Arial];
list / style={liststyletype="decimal" fontsize=11pt};
item "{Statistical Methods, Page} {\field {\*\fldinst PAGEREF Methods \h}}
^n ^n"/style=[fontsize=11pt fontfamily=Arial color=blue url='#Methods'];
...
item "{Boxplot of Age by Cholesterol Status, Page} {\field {\*\fldinst
PAGEREF Figure1 \h}}^n ^n"/style=[fontsize=11pt fontfamily=Arial color=blue
url='#Figure1'];
end;
run;
```

SAS Syntax	RTF/SAS Function
{ }	Encloses section. Text can be inserted prior to the field call using brackets
PAGEREF	Tells RTF to reference page number of bookmark
\\h	Creates hyperlink for page number

Table 6. Explanation of bookmark page insertion syntax

Once the RTF is output, one must use CTRL+A and F9 to update the bookmark location similar to the automatic TOC above.

Table of Contents
1. Statistical Methods, Page 2
2. Patient and Baseline Descriptive Characteristics Overall and by Gender, Page 3
3. Association between Age and Cholesterol, Page 4
4. Boxplot of Age by Cholesterol Status, Page 5

Output 9. Manual Table of Contents with Page Numbers

PROS AND CONS:

PROS:	CONS:
<ul style="list-style-type: none"> • Customizable in SAS 	<ul style="list-style-type: none"> • Subject to typos and incorrect link specification
<ul style="list-style-type: none"> • Reproducible 	<ul style="list-style-type: none"> • Need to use commands in Word to update page numbers
<ul style="list-style-type: none"> • Can move in both directions throughout document 	
<ul style="list-style-type: none"> • Automatic page number insertion 	

Table 7. Pros and Cons of using PROC ODSTEXT and RTF ANCHORS

In contrast to a single automatic TOC or use of ODS RTF ANCHORS, ODS RTF BOOKMARKS can be formatted seamlessly in SAS and will work throughout the report! Further, page numbers are stored within Microsoft Word if one wants to add these. Unfortunately, one will still need to run two shortcut keys to get the page numbers to appear.

CONCLUSION

As with most aspects of SAS, multiple methods exist for creating a unique table of contents and for guiding readers through a report. This paper presents a few ways to help readers follow along in reports with Table of Contents and links throughout the document. The trade-off between automatic and manual is that of convenience versus customization. However, one doesn't necessarily have to choose, as these methods can be combined in a single report for different uses.

REFERENCES

1. SAS Institute Inc. 2016. *Sashelp Data Sets*. Cary, NC: SAS Institute Inc.
2. Parsons, L. S. 2007. "Enhancing RTF Output with RTF Control Words and In-Line Formatting". *Proceedings of the SAS Global Forum 2007*. Orlando, FL. SAS Institute Inc., Available at: <http://www2.sas.com/proceedings/forum2007/151-2007.pdf>
3. Parsons, L. S. 2008. "Creating a Custom Table of Contents in RTF Documents". *Proceedings of the SAS Global Forum 2008*. San Antonio, TX. SAS Institute Inc., Available at: <http://www2.sas.com/proceedings/forum2008/238-2008.pdf>
4. Khurshed, F. 2011. "Creating a Table of Contents for MS Word using AutoFormat and Cascading Style Sheets". *Proceedings of the SAS Global Forum 2011*. Las Vegas, NV. SAS Institute Inc., Available at: https://www.sas.com/content/dam/SAS/en_ca/User%20Group%20Presentations/Vancouver-User-Group/FareezaKhurshed-CreatingTOCforWord-2011.pdf
5. Khurshed, F. 2011. "Creating a Table of Contents for MS Word using AutoFormat and Cascading Style Sheets". *Proceedings of the SAS Global Forum 2011*. Las Vegas, NV. SAS Institute Inc., Available at: https://www.sas.com/content/dam/SAS/en_ca/User%20Group%20Presentations/Vancouver-User-Group/FareezaKhurshed-CreatingTOCforWord-2011.pdf
6. SAS Institute Inc., 2004. "Usage Note 24050: How to create a table of contents with ODS RTF". Accessed July 1, 2018: Available at: <http://support.sas.com/kb/24/050.html>.
7. Goldman, D. A. 2017. "Creating Complete Automated Reports in SAS using the ODS RTF Destination: Protect your Results from Human Error". *Proceedings of the Western Users of SAS Software Conference 2017*. Long Beach, CA. SAS Institute Inc., Available at: https://www.lexjansen.com/wuss/2017/32_Final_Paper_PDF.pdf
8. SAS Institute Inc. 2011. "SAS 9 ODS List and Text Block Tip Sheet". Accessed July 1, 2018. Available at: http://support.sas.com/rnd/base/ods/Tipsheet_ListTextBlks.pdf
9. Osowski, S., Fritchey, T. 2006. "Hyperlinks and Bookmarks with ODS RTF". *Proceedings of the Pharmaceutical SAS Users Group 2006*. SAS Institute Inc., Available at: <https://www.lexjansen.com/pharmasug/2006/TechnicalTechniques/TT21.pdf>

ACKNOWLEDGMENTS

Thank you to Kaitlin Woo, MS, for reviewing and providing feedback for this paper.

RECOMMENDED READING

- SAS® *Output Delivery System: Procedures Guide, Third Edition*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Debra A. Goldman
Memorial Sloan Kettering Cancer Center
485 Lexington Avenue 2nd Floor
New York, NY 10022
646-888-8331
goldmand@mskcc.org
<https://www.mskcc.org/profile/debra-goldman>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.